

Tema 10

Funcions de diverses variables. Límits, continuïtat i diferenciabilitat.

Objectius:

1. Definir funcions de diverses variables amb wxMaxima. Calcular amb funcions de diverses variables.
2. Representar funcions de dues variables gràficament. Practicar amb les diverses opcions.
3. Calcular límits de funcions reals de diverses variables.
4. Estudiar la continuïtat de funcions de diverses variables.
5. Calcular derivades parcials de funcions reals de diverses variables
6. Calcular matrius jacobianes y estudiar les propietats.
7. Calcular derivades d'ordre superior de funcions de diverses variables.

Continguts:

- 10-1. Funcions de diverses variables.
- 10-2. Representació gràfica de funcions reals de dues variables.
- 10-3. Límits de funcions reals de dues variables. Continuïtat.
- 10-4. Derivades parcials. Derivades direccionals. Derivades d'ordre superior.
- 10-5. Matriu jacobiana. Diferenciabilitat. La regla de la cadena.

Referències

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.
- APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.

- AP86 APOSTOL, T.M. (1986)
Análisis Matemático
- BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR,
JOSÉ (2009)
Modelos matemáticos con Maxima
- ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.
- RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J.
RAFAEL (2005)
Introducción a Maxima
- RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería
- RU80 RUDIN, WALTER (1980)
Principios de Análisis Matemático.
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

10-1.- Funcions de diverses variables.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_10-1.wxm**.

10.1.1 Funcions reals de diverses variables.

S'anomena funció real de diverses variables o camp escalar, qualsevol terna (A, B, f) en la qual $A \subset \mathbb{R}^n, B \subset \mathbb{R}$ i f és una correspondència entre els dos conjunts, tal que per a tot element $x \in A$ existeix un únic element assignat $y \in B$. Per expressar una funció real de diverses variables és habitual escriure una equació de la forma $y = f(x_1, x_2, \dots, x_n), x \in A \subset \mathbb{R}^n$.

Notació:

- Si $n = 2$, s'acostuma a escriure $x_1 = x, x_2 = y$.
- Si $n = 3$, s'acostuma a escriure $x_1 = x, x_2 = y, x_3 = z$.

Per calcular el domini o camp d'existència d'una funció real de diverses variables, cal establir el major subconjunt en el qual té sentit l'expressió que determina el valor de la imatge.

Vegem com es defineixen les funcions reals de diverses variables amb wxMaxima mitjançant alguns exemples.

Exemple 10.1.1.

(a) Es considera la funció definida per: $f(x, y) = 2(x-1)^2 + 2(y-1)^2$

Per definir-la se segueix una sintaxi molt similar al cas de funcions reals de variable real:

```
(%i1) f1(x,y):=2*(x-1)^2+2*(y-1)^2;  
(%o1) f1(x,y):= 2 (x-1)^2 + 2 (y-1)^2
```

Per calcular el valor de la funció en un punt, senzillament:

```
(%i2) 'f1(2,3)=f1(2,3);  
(%o2) f1(2,3)= 10
```

(b) El volum d'un cilindre recte circular és un número real positiu que depèn del radi de la base (r) i de l'altura (h) i, per tant, es pot expressar mitjançant una funció real de dues variables: $V(r, h) = \pi r^2 h, r, h > 0$.

Aleshores:

```
(%i3) V(r,h):=%pi*r^2*h;
(%o3) V(r,h):=  $\pi r^2 h$ 
(%i4) 'V(5,7)=V(5,7);
(%o4) V(5,7)=  $175 \pi$ 
```

(c) Funció real de tres variables

```
(%i5) f2(x,y,z):=(x+y)^2+exp(z);
      'f2(1,2,3)=f2(1,2,3);
(%o5) f2(x,y,z):=  $(x+y)^2 + \exp(z)$ 
(%o6) f2(1,2,3)=  $e^3 + 9$ 
```

(d) Funció real de quatre variables:

```
(%i7) f3(x,y,u,v):=x^2+x^3*y-2*u^2+exp(v);
      'f3(1,-1,1,2)=f3(1,-1,1,2);
(%o7) f3(x,y,u,v):=  $x^2 + x^3 y + (-2) u^2 + \exp(v)$ 
(%o8) f3(1,-1,1,2)=  $e^2 - 2$ 
```

10.1.2.- Operacions amb funcions reals de diverses variables

Es poden dur a terme operacions amb funcions reals de diverses variables. Considerem per exemple les funcions reals de dues variables:

$$f_1(x,y) = 2x^2 + y^2 + 1; \quad f_2(x,y) = x^2 \sin(x+y)$$

Definició de les funcions amb wxMaxima:

```
(%i1) f1(x,y):=2*x^2+y^2+1;    f2(x,y):=x^2*sin(x+y);
(%o1) f1(x,y):=  $2x^2 + y^2 + 1$ 
(%o2) f2(x,y):=  $x^2 \sin(x+y)$ 
```

Suma:

```
(%i3) '(f1(x,y)+f2(x,y))=f1(x,y)+f2(x,y);
(%o3) f2(x,y)+f1(x,y)=  $x^2 \sin(y+x) + y^2 + 2x^2 + 1$ 
```

Producte per un escalar:

```
(%i4) '(a*f1(x,y))=a*f1(x,y);
(%o4) a f1(x,y)=a (y^2+2 x^2+1)
(%i5) expand(%);
(%o5) a f1(x,y)=a y^2+2 a x^2+a
```

Producte:

```
(%i7) '(f1(x,y)*f2(x,y))=f1(x,y)*f2(x,y);
(%o7) f1(x,y) f2(x,y)=x^2 (y^2+2 x^2+1) sin(y+x)
(%i8) expand(%);
(%o8) f1(x,y) f2(x,y)=x^2 y^2 sin(y+x)+2 x^4 sin(y+x)+x^2 sin(y+x)
```

Quocient:

```
(%i9) '(f1(x,y)/f2(x,y))=f1(x,y)/f2(x,y);
(%o9)  $\frac{f1(x,y)}{f2(x,y)} = \frac{y^2+2 x^2+1}{x^2 \sin(y+x)}$ 
```

10.1.3.- Funcions vectorials de diverses variables.

S'anomena funció vectorial de diverses variables o camp vectorial, qualsevol terna (A, B, f) en la qual $A \subset \mathbb{R}^n, B \subset \mathbb{R}^m$ i f és una correspondència entre els dos conjunts, tal que per a tot element $x \in A$ existeix un únic element assignat $y \in B$.

Per expressar una funció vectorial de diverses variables és habitual escriure un vector definit mitjançant les components de la funció, que són funcions reals de diverses variables. Així, doncs:

$$f(x_1, x_2, \dots, x_n) = (f_1(x_1, x_2, \dots, x_n), \dots, f_m(x_1, x_2, \dots, x_n)), x \in A \subset \mathbb{R}^n$$

Vegem la definició amb wxMaxima mitjançant un exemple:

```
(%i1) f1(x,y,z):=x^2+y-sin(z);      f2(x,y,z):=z*exp(x+y);
      f(x,y,z):=[f1(x,y,z),f2(x,y,z)];
(%o1) f1(x,y,z):=x^2+y-sin(z)
(%o2) f2(x,y,z):=z exp(x+y)
(%o3) f(x,y,z):=[f1(x,y,z),f2(x,y,z)]

(%i4) 'f(1,2,-1)=f(1,2,-1);
(%o4) f(1,2,-1)=[sin(1)+3,-%e^3]
```

10.1.4.- Operacions amb funcions vectorials de diverses variables

Si es consideren dues funcions:

$$f: A_1 \subset \mathbb{R}^n \rightarrow \mathbb{R}^m, \quad g: A_2 \subset \mathbb{R}^n \rightarrow \mathbb{R}^m$$

es poden dur a terme component a component les operacions suma i producte per un escalar real.

Suma:

```
(%i1) f1(x,y,z):=x^2+y-sin(z);      f2(x,y,z):=z*exp(x+y);  
      f(x,y,z):=[f1(x,y,z),f2(x,y,z)];  
(%o1) f1(x,y,z):=x^2+y-sin(z)  
(%o2) f2(x,y,z):=z exp(x+y)  
(%o3) f(x,y,z):=[f1(x,y,z),f2(x,y,z)]  
  
(%i4) g1(x,y,z):=2*x^2+y^2+3*z+1;   g2(x,y,z):=z^2*sin(x+y);  
      g(x,y,z):=[g1(x,y,z),g2(x,y,z)];  
(%o4) g1(x,y,z):=2 x^2+y^2+3 z+1  
(%o5) g2(x,y,z):=z^2 sin(x+y)  
(%o6) g(x,y,z):=[g1(x,y,z),g2(x,y,z)]  
  
(%i7) '((f+g)(x,y,z))=f(x,y,z) + g(x,y,z);  
(%o7) (g+f)(x,y,z)=[-sin(z)+3 z+y^2+y+3 x^2+1, sin(y+x) z^2+%e^{y+x} z]
```

Producte per un escalar real:

```
(%i8) '((af)(x,y,z))=a*f(x,y,z);  
(%o8) af(x,y,z)=[a (-sin(z)+y+x^2), a %e^{y+x} z]
```

Producte escalar:

```
(%i9) f(x,y,z).g(x,y,z);  
(%o9) (3 z+y^2+2 x^2+1)(-sin(z)+y+x^2)+%e^{y+x} sin(y+x) z^3
```

Si es consideren dues funcions vectorials

$$f: A_1 \subset \mathbb{R}^n \rightarrow \mathbb{R}^m, \quad g: A_2 \subset \mathbb{R}^m \rightarrow \mathbb{R}^p$$

es defineix la funció composta $g \circ f$ mitjançant:

$$(g \circ f)(x) = g(f(x)); \quad \text{Dom}(g \circ f) = \{x \in A_1, f(x) \in A_2\}$$

Per exemple:

```
(%i1) f1(r,a):=r*cos(theta);      f2(r,a):=r*sin(theta);
      g(x,y):=sqrt(R^2 - x^2 - y^2);
      F(r,a)=g(f1(r,theta), f2(r,theta));
(%o1) f1(r,a):=r cos(theta)
(%o2) f2(r,a):=r sin(theta)
(%o3) g(x,y):=sqrt(R^2-x^2-y^2)
(%o4) F(r,a)=sqrt(R^2-r^2 sin(theta)^2-r^2 cos(theta)^2)

(%i5) trigsimp(%);
(%o5) F(r,a)=sqrt(R^2-r^2)
```

Vegem un altre exemple:

```
(%i1) g(x,y,z):=x*y^2+exp(z);
      f1(t):=2*t+1;      f2(t):=t^2+2;      f3(t):=sin(t);
      f(t):=[f1(t), f2(t), f3(t)];
(%o1) g(x,y,z):=x y^2+exp(z)
(%o2) f1(t):=2 t+1
(%o3) f2(t):=t^2+2
(%o4) f3(t):=sin(t)
(%o5) f(t):=[f1(t),f2(t),f3(t)]

(%i6) '(g(f(t)))=g(f1(t),f2(t),f3(t));
(%o6) g(f(t))=%e^sin(t)+(2 t+1)(t^2+2)^2
```

10-2.- Representació gràfica de funcions reals de dues variables.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_10-2.wxm**.

Per representar gràficament una funció real de dues variables cal:


- 1) Definir la funció
- 2) Aplicar les opcions del menú “Gràficos 3D”, seleccionant les opcions que més interessin per a la representació gràfica que es vol fer.

El quadre de diàleg de Gràficos 3D és:



Breu descripció de les opcions del quadre:

- Expresión: cal escriure la notació assignada a la funció.
- Variable x: cal especificar el rang de la variable x.
- Variable y: cal especificar el rang de la variable y.
- Cuadrícula: opció per modificar la graduació de la malla en la qual es fa la gràfica.
- Formato: hi ha quatre opcions: predeterminado, en línea, gnuplot i openmath. Fan referència a quatre formats de sortida de la gràfica que es veuran als Exemples.



- Opciones: hi ha sis opcions que configuren la forma de la gràfica, tal com es veurà als Exemples.



Exemple 10.2.1. Es farà la representació gràfica en diferents formats de la funció


$$f_1(x, y) = 2(x-1)^2 + 2(y-1)^2$$

Definició de la funció:

(%i1) f2(x,y):=x^2-y^2;

(%o1) f2(x,y):=x^2-y^2


Gràfic amb el format “predeterminado” havent fet una rotació de la gràfica obtinguda:



Gràfic amb el format “en línea” i diferents opcions


```
(%i3) wxplot3d(f1(x,y), [x,-5,5], [y,-5,5],  
[gnuplot_preamble, "set pm3d at b"])$
```

(%t3)




```
(%i4) wxplot3d(f1(x,y), [x,-5,5], [y,-5,5],  
[gnuplot_preamble, "set hidden3d"])$
```

(%t4)



```
(%i5) wxplot3d(f1(x,y), [x,-5,5], [y,-5,5],  
[gnuplot_preamble, "set pm3d at s; unset surf; unset colorbox"])$
```

(%t5)



Exemple 10.2.2. Es farà la representació gràfica en diferents formats de la funció


$$f_1(x, y) = x^2 - y^2$$

Definició de la funció:

```
(%i1) f1(x,y):=2*(x-1)^2+2*(y-1)^2;
(%o1) f1(x,y):= 2 (x-1)^2 + 2 (y-1)^2
```


El format “predeterminado” obre una finestra que permet veure la gràfica amb diferents angles i perspectives. Es recomana aplicar aquesta opció en el fitxer d’aquest apartat. Amb l’opció “en línea” el gràfic queda “enganxat” al propi fitxer de càlculs.

Gràfic amb el format “predeterminado” havent fet una rotació de la gràfica obtinguda:




Gràfic amb el format “en línea” i diferents opcions

```
(%i3) wxplot3d(f2(x,y), [x,-5,5], [y,-5,5],
[gnuplot_preamble, "set pm3d at b"])$
```




```
(%i4) wxplot3d(f2(x,y), [x,-5,5], [y,-5,5],
[gnuplot_preamble, "set hidden3d"])$
```

(%t4)



```
(%i5) wxplot3d(f2(x,y), [x,-5,5], [y,-5,5],
[gnuplot_preamble, "set pm3d at s; unset surf; unset colorbox"])$
```

(%t5)



10-3.- Límits de funcions reals de dues variables. Continuïtat

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_10-3.wxm**.

Límit en un punt d'una funció real de dues variables.

Si $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ i (x_0, y_0) és un punt d'acumulació del camp d'existència de la funció, es diu que la funció té límit $L \in \mathbb{R}$ en el punt (x_0, y_0) si qualsevol que sigui $\varepsilon > 0$ existeix $\delta > 0$ tal que per a tot $(x, y) \in (B^*((x_0, y_0); \delta)) \cap A$ es compleix $|f(x, y) - L| < \varepsilon$. Notació: $L = \lim_{(x, y) \rightarrow (x_0, y_0)} f(x, y)$.

El límit d'una funció en un punt, si existeix, és únic.

Límit en un punt d'una funció real de dues variables segons un subconjunt

Si $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$, $(x_0, y_0) \in A'$ i $W \subset \mathbb{R}^2$, es diu que la funció té límit $L \in \mathbb{R}$ en el punt (x_0, y_0) segons el subconjunt W , si qualsevol que sigui $\varepsilon > 0$ existeix $\delta > 0$ tal que per a tot $(x, y) \in (B^*((x_0, y_0); \delta)) \cap W$ es compleix $|f(x, y) - L| < \varepsilon$.

Notació:

$$L = \lim_{\substack{(x, y) \rightarrow (x_0, y_0) \\ W}} f \quad \text{o be} \quad L = \lim_{\substack{(x, y) \rightarrow (x_0, y_0) \\ (x, y) \in W}} f(x, y)$$

Si el subconjunt W és una recta, el límit s'anomena direccional.

Si existeixen dos subconjunts W_1, W_2 tals que el límit respectiu de la funció en el punt segons cada un d'ells és diferent, aleshores el límit de la funció en el punt no existeix.

En wxMaxima no hi ha cap instrucció implementada per calcular el límit de funcions de dues ni de més variables. Per tant, cal aplicar els criteris a aquest efecte aplicant els càlculs amb el programa.

Exemple 10.3.1. Es considera la funció:

$$f(x, y) = \frac{2x^2y}{2x^2 + y^2}, \quad \text{si } (x, y) \neq (0, 0)$$

El domini d'aquesta funció és $A = \mathbb{R}^2 - \{(0, 0)\}$ i el punt $(0, 0)$ és un punt d'acumulació de A . Verifiquem que $\lim_{(0, 0)} f = 0$. En efecte:

$$|f(x, y) - 0| = \left| \frac{2x^2y}{2x^2 + y^2} \right| = \frac{2x^2}{2x^2 + y^2} |y| \leq |y| < \varepsilon$$

sempre que $|(x, y) - (0, 0)| = \sqrt{x^2 + y^2} < \delta$, amb $\delta \leq \varepsilon$.

Exemple 10.3.2. Es considera la funció:

$$f(x, y) = \frac{x^2}{x^2 + y^2}, \quad \text{si } (x, y) \neq (0, 0)$$

El domini d'aquesta funció és $A = \mathbb{R}^2 - \{(0, 0)\}$ i el punt $(0, 0)$ és un punt d'acumulació de A . Si es considera el subconjunt $W = \{(x, y) : y = mx\}$ que és una recta que passa per l'origen, aleshores el límit (direccional) de la funció segons aquest subconjunt en el punt és:

```
(%i1) f(x,y):=(x^2)/(x^2+y^2);
```

```
(%o1) f(x,y):=  $\frac{x^2}{x^2+y^2}$ 
```

```
(%i2) f(x,m*x);
```

```
(%o2)  $\frac{x^2}{m^2 x^2 + x^2}$ 
```

```
(%i3) L1:=limit( f(x,m*x) , x, 0);
```

```
(%o3) L1 =  $\frac{1}{m^2 + 1}$ 
```

Per tant el límit direccional depèn de la recta i en conclusió, el límit de la funció en el punt $(0, 0)$ no existeix.

Límits reiterats. Si $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ i $(x_0, y_0) \in A'$, s'anomenen límits reiterats de la funció en aquest punt, els límits següents:

$$L_{xy} = \lim_{y \rightarrow y_0} \left(\lim_{x \rightarrow x_0} f(x, y) \right); \quad L_{yx} = \lim_{x \rightarrow x_0} \left(\lim_{y \rightarrow y_0} f(x, y) \right)$$

Càlcul amb wxMaxima:

```
(%i1) limit ( limit(f(x,y), y, y0) , x, x0);
```

```
(%o1)  $\lim_{x \rightarrow x_0} \lim_{y \rightarrow y_0} f(x, y)$ 
```

```
(%i2) limit ( limit(f(x,y), x, x0) , y, y0);
(%o2) lim lim f(x,y)
      y→y0 x→x0
```

Propietats dels límits reiterats:

- Si existeix el límit de la funció $L = \lim_{(x_0, y_0)} f$, aleshores qualsevol dels límits reiterats que existeixi ha de tenir aquest valor.
- Pot ser que algun dels límits reiterats o ambdós no existeixin i sí que existeixi el límit de la funció $L = \lim_{(x_0, y_0)} f$.
- Si els límits reiterats existeixen però són diferents, aleshores el límit de la funció en el punt no existeix.
- Si els límits reiterats existeixen i són iguals, això no implica que el límit de la funció en el punt existeixi.

Exemple 10.3.3. Es considera la funció definida per

$$f(x, y) = \frac{x^2 y}{x^4 + y^2}, \quad \text{si } (x, y) \neq (0, 0).$$

Aleshores es compleix:

```
(%i1) f(x,y):=(x^2 * y)/(x^4+y^2);
(%o1) f(x,y):= \frac{x^2 y}{x^4 + y^2}
```

Límits reiterats a l'origen:

```
(%i2) 'Lyx(0,0) = limit ( limit(f(x,y), y, 0) , x, 0);
(%o2) Lyx(0,0)=0

(%i3) 'Lxy(0,0) = limit ( limit(f(x,y), x, 0) , y, 0);
(%o3) Lxy(0,0)=0
```

Així, doncs, els límits reiterats a l'origen existeixen i són iguals. Tanmateix calculem el límit direccional segons les paràboles d'equació $y = x^2$:

```
(%i4) 'L[y=x^2](0,0) = limit(f(x,x^2), x, 0);
(%o4) L_{y=x^2}(0,0)=\frac{1}{2}
```

Per tant, el límit de la funció a l'origen no existeix.

Exemple 10.3.4. Es considera la funció definida per:

$$f(x, y) = \begin{cases} (x + y) \sin\left(\frac{1}{x}\right) \sin\left(\frac{1}{y}\right), & \text{si } (x, y) \neq (0, 0) \\ 0, & \text{si } x = 0 \text{ o } y = 0 \end{cases}.$$

Aleshores es compleix:

$$\begin{aligned} (\%i1) \quad & f(x, y) := (x + y) * \sin(1/x) * \sin(1/y); \\ (\%o1) \quad & f(x, y) := (x + y) \sin\left(\frac{1}{x}\right) \sin\left(\frac{1}{y}\right) \end{aligned}$$

Límits reiterats a l'origen:

$$\begin{aligned} (\%i2) \quad & 'Lyx(0, 0) = \text{limit}(\text{limit}(f(x, y), y, 0), x, 0); \\ (\%o2) \quad & Lyx(0, 0) = \text{ind} \\ (\%i3) \quad & 'Lxy(0, 0) = \text{limit}(\text{limit}(f(x, y), x, 0), y, 0); \\ (\%o3) \quad & Lxy(0, 0) = \text{ind} \end{aligned}$$

Així, doncs, els límits reiterats a l'origen no existeixen. Tanmateix el límit de la funció a l'origen existeix i val 0 ja que és el producte d'una funció de límit 0 per una funció fitada.

El Teorema de descomposició del domini.

Sigui $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ i (x_0, y_0) un punt d'acumulació del domini. Si el domini de la funció es descompon en reunió finita de subconjunts $A = W_1 \cup \dots \cup W_p$ i es compleix que $L = \lim_{(x_0, y_0) \in W_k} f$, $k = 1, 2, \dots, p$ aleshores es verifica que $\lim_{(x_0, y_0)} f = L$.

Exemple 10.3.5. Per calcular el límit $\lim_{(0,0)} \frac{e^{xy} - 1}{\sqrt{x^2 + y^2}}$ aplicarem el teorema de descomposició del domini, de la manera següent:

$$\begin{aligned} A = \mathbb{R}^2 - \{(0, 0)\} &= W_1 \cup W_2 \cup W_3 = \\ &= \{(x, y) \in \mathbb{R}^2 : x = 0, y \neq 0\} \cup \{(x, y) \in \mathbb{R}^2 : x \neq 0, y = 0\} \cup \{(x, y) \in \mathbb{R}^2 : x \neq 0, y \neq 0\} \end{aligned}$$

verificant-se que:

```
(%i1) f(x,y):=(%e^(x*y) - 1) / sqrt(x^2 + y^2) ;
```

```
(%o1) f(x,y):=
$$\frac{e^{xy} - 1}{\sqrt{x^2 + y^2}}$$

```

```
(%i2) 'LW1(0,0) = limit(f(0,y), x, 0);
```

```
(%o2) LW1(0,0)=0
```

```
(%i3) 'LW2(0,0) = limit(f(x,0), y, 0);
```

```
(%o3) LW2(0,0)=0
```

El límit segons el tercer subconjunt es calcula mitjançant:

$$\lim_{\substack{(0,0) \\ W_3}} \frac{e^{xy} - 1}{xy} \cdot \frac{xy}{\sqrt{x^2 + y^2}} = \lim_{\substack{(0,0) \\ W_3}} \frac{e^{xy} - 1}{xy} \cdot \frac{x}{\sqrt{x^2 + y^2}} y = 1 \cdot \text{fitada} \cdot 0 = 0$$

Per tant, es compleix que $\lim_{(0,0)} \frac{e^{xy} - 1}{\sqrt{x^2 + y^2}} = 0$.

Continuïtat de funcions reals de dues variables

Si $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ i $(x_0, y_0) \in A$, és un punt del domini que també és d'acumulació (és a dir és un punt no aïllat del domini), es diu que la funció és continua en aquest punt si es compleix

$$\lim_{(x,y) \rightarrow (x_0,y_0)} f(x,y) = f(x_0,y_0).$$

Si en un punt una funció no és continua, es diu que és *discontinua*. Es diu que f és continua en $X \subset A$ si la funció és continua en tots els punts d'aquest subconjunt. El major subconjunt del domini en que una funció és continua s'anomena camp de continuïtat de la funció.

10-4.- Derivades parcials. Derivades direccionals. Derivades d'ordre superior.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_10-4.wxm**.

10.4.1. Derivades parcials de funcions reals de dues variables

Considerem una funció $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ i un punt $a = (a_1, a_2) \in \overset{\circ}{A}$. Es defineixen les derivades parcials de f en el punt a mitjançant:

$$D_x f(a) = \lim_{t \rightarrow 0} \frac{f(a_1 + t, a_2) - f(a_1, a_2)}{t} ; \quad D_y f(a) = \lim_{t \rightarrow 0} \frac{f(a_1, a_2 + t) - f(a_1, a_2)}{t}$$

Notació:

$$D_x f(a) \equiv D_1 f(a) \equiv \frac{\partial f}{\partial x}(a) ; \quad D_y f(a) \equiv D_2 f(a) \equiv \frac{\partial f}{\partial y}(a)$$

Una funció $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ es diu derivable parcialment en $a \in \overset{\circ}{A}$, si en aquest punt existeixen les dues derivades parcials de la funció.

Una funció $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ es diu derivable parcialment en un subconjunt del domini si existeixen les dues derivades parcials de la funció en tots els punts d'aquest conjunt. Les funcions que assignen a cada punt la derivada parcial corresponent s'anomenen funcions derivades parcials de f .

Vegem com es calculen les derivades parcials amb wxMaxima.

Exemple 10.4.1. Si es considera la funció definida per $f(x, y) = 4x^2 + 3y^2$, es tracta de calcular les funcions derivades parcials i les derivades parcials en el punt $a = (1, 2)$.

Primer cal definir la funció:

```
(%i1) f(x,y):=4*x^2 + 3*y^2;  
(%o1) f(x,y):= 4 x^2 + 3 y^2
```

A continuació cal aplicar les instruccions de càlcul de derivades, anàlogues al cas de funcions d'una variable, però en aquest cas s'ha d'especificar la variable respecte a la qual es deriva. Es pot obtenir l'expressió de la funció derivada parcial corresponent amb la instrucció "define":

```
(%i2) define (Dxf(x,y) , diff(f(x,y), x) );
      define (Dyf(x,y) , diff(f(x,y), y) );
(%o2) Dxf(x,y):= 8 x
      (%o3) Dyf(x,y):= 6 y
```

I finalment es pot calcular el valor de les funcions derivades parcials al punt corresponent:

```
(%i4) 'Dxf(1,2)=Dxf(1,2);
      'Dyf(1,2)=Dyf(1,2);
(%o4) Dxf(1,2)= 8
      (%o5) Dyf(1,2)= 12
```

En alguns casos el càlcul de la derivada parcial en un punt no es pot dur a terme d'una forma tant senzilla i cal recórrer a la definició per tal de verificar si la funció és o no derivable parcialment en el punt, tal com es mostra a l'Exemple que segueix.

Exemple 10.4.2. E considera la funció definida per

$$f(x,y) = \begin{cases} \frac{x^3 + y^3}{x^2 + y^2}, & \text{si } (x,y) \neq (0,0) \\ 0, & \text{si } (x,y) = (0,0) \end{cases}.$$

Es tracta de calcular les funcions derivades parcials, les derivades parcials en un punt qualsevol $(x_0, y_0) \neq (0,0)$ i les derivades parcials en el punt $(0,0)$. Comencem calculant les derivades parcials d'aquesta funció:

```
(%i1) f(x,y):=(x^3 + y^3)/(x^2 + y^2);
(%o1) f(x,y):= \frac{x^3+y^3}{x^2+y^2}

(%i2) define (Dxf(x,y) , diff(f(x,y), x) );
      define (Dyf(x,y) , diff(f(x,y), y) );
(%o2) Dxf(x,y):= \frac{3 x^2}{y^2+x^2} - \frac{2 x (y^3+x^3)}{(y^2+x^2)^2}
      (%o3) Dyf(x,y):= \frac{3 y^2}{y^2+x^2} - \frac{2 y (y^3+x^3)}{(y^2+x^2)^2}
```

Ara ja podem calcular les derivades parcials en un punt qualsevol $(x_0, y_0) \neq (0,0)$:

```
(%i4) 'Dxf(x0,y0) = Dxf(x0,y0);    'Dyf(x0,y0) = Dyf(x0,y0);
(%o4) Dxf(x0,y0) =  $\frac{3x_0^2}{y_0^2+x_0^2} - \frac{2x_0(y_0^3+x_0^3)}{(y_0^2+x_0^2)^2}$ 
(%o5) Dyf(x0,y0) =  $\frac{3y_0^2}{y_0^2+x_0^2} - \frac{2y_0(y_0^3+x_0^3)}{(y_0^2+x_0^2)^2}$ 
```

Les derivades a l'origen no es poden calcular aplicant les funcions derivades parcials en aquest punt. En efecte:

```
(%i6) 'Dxf(0,0)=Dxf(0,0);    'Dyf(0,0)=Dyf(0,0);
Division by 0
#0: Dxf(x=0,y=0)
-- an error. To debug this try: debugmode(true);
Division by 0
#0: Dyf(x=0,y=0)
-- an error. To debug this try: debugmode(true);
```

Per tant s'ha d'aplicar la definició de derivada parcial:

```
(%i8) 'Dxf(0,0) = limit( (f(t,0) - 0)/(t) , t, 0);
(%o8) Dxf(0,0)=1

(%i9) 'Dyf(0,0) =limit( (f(0,t) - 0)/(t) , t, 0);
(%o9) Dyf(0,0)=1
```

Així doncs la funció és derivable parcialment a l'origen i s'ha calculat el valor de cada una d'aquestes derivades parcials.

10.4.2 Derivades direccionals de funcions reals de dues variables

Considerem una funció $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ i un punt $a \in \overset{\circ}{A}$. Es defineix la derivada de f en aquest punt en la direcció del vector unitari $u \in \mathbb{R}^2$ com el límit, si existeix:

$$D_u f(a) = \lim_{t \rightarrow 0} \frac{f(a + tu) - f(a)}{t}$$

Notació: $D_u f(a) \equiv \frac{\partial f}{\partial u}(a)$.

Exemple 10.4.3. E considera la funció definida per $f(x, y) = x + 2y$; es vol calcular la derivada direccional segons el vector $v = (1, 1)$ en el punt $a = (3, 2)$.

En primer lloc s'observa que el vector donat no és unitari i, per tant, cal normalitzar-lo:

```
(%i1) 'u=u:[1,1];      modul:sqrt(1^2+1^2);      'u[1]=u/modul;
(%o1) u = [ 1, 1]
(%o2) sqrt(2)
(%o3) u_1 = [ 1/sqrt(2), 1/sqrt(2)]
```

Ara definim la funció:

```
(%i4) f(x,y):=x+2*y;
(%o4) f(x,y):=x+2 y
```

I procedim al càlcul de la derivada direccional:

```
(%i5) 'Duf(3,2) = limit(( f(3+t/sqrt(2),2+t/sqrt(2)) - f(3,2))/t, t, 0);
(%o5) Duf(3,2)=sqrt(2)+1/sqrt(2)
```

10.4.3. Derivades d'ordre superior de funcions reals de dues variables

Considerem una funció $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$. Si la funció és derivable parcialment en un subconjunt $\Omega \subset A$, existeixen les derivades parcials $D_1 f, D_2 f$ definides en Ω . Si aquestes funcions són derivables parcialment, les seves derivades parcials s'anomenen derivades parcials de segon ordre.

Per a cada $a = (a_1, a_2) \in \Omega$ es defineixen així:

$$\begin{aligned} D_{xx}f(a) &= D_x(D_x f)(a) = \lim_{t \rightarrow 0} \frac{D_x f(a_1 + t, a_2) - D_x f(a_1, a_2)}{t} \\ D_{xy}f(a) &= D_y(D_x f)(a) = \lim_{t \rightarrow 0} \frac{D_x f(a_1, a_2 + t) - D_x f(a_1, a_2)}{t} \\ D_{yx}f(a) &= D_x(D_y f)(a) = \lim_{t \rightarrow 0} \frac{D_y f(a_1 + t, a_2) - D_y f(a_1, a_2)}{t} \\ D_{yy}f(a) &= D_y(D_y f)(a) = \lim_{t \rightarrow 0} \frac{D_y f(a_1, a_2 + t) - D_y f(a_1, a_2)}{t} \end{aligned}$$

Notació:

$$D_{xx}f(a) \equiv D_{11}f(a) \equiv \frac{\partial^2 f}{\partial x^2}(a) \quad ; \quad D_{xy}f(a) \equiv D_{12}f(a) \equiv \frac{\partial^2 f}{\partial x \partial y}(a)$$

$$D_{yx}f(a) \equiv D_{21}f(a) \equiv \frac{\partial^2 f}{\partial y \partial x}(a) \quad ; \quad D_{yy}f(a) \equiv D_{22}f(a) \equiv \frac{\partial^2 f}{\partial y^2}(a)$$

Metodologia de càlcul: atès que es tracta de derivar parcialment, s'aplica la mateixa metodologia que s'ha comentat per a les derivades parcials de primer ordre.

Les derivades $D_{xy}f(x, y)$ i $D_{yx}f(x, y)$, s'anomenen derivades creuades (o mixtes). Per aquestes derivades hi ha un teorema important.

Teorema (Schwarz). Si una funció $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ és tal que en un entorn del punt $(a, b) \in A$ existeixen les derivades parcials $D_x f(x, y)$, $D_y f(x, y)$, $D_{xy}f(x, y)$ i aquesta derivada parcial de segon ordre és contínua en el punt $(a, b) \in A$, aleshores existeix la derivada de segon ordre $D_{yx}f(a, b)$ i es compleix la igualtat $D_{yx}f(a, b) = D_{xy}f(a, b)$.

La matriu hessiana. Si una funció $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ és derivable dues vegades en un punt $(a, b) \in A$, s'anomena matriu hessiana de la funció en aquest punt, la matriu quadrada

$$Hf(a, b) = \begin{pmatrix} D_{xx}f(a, b) & D_{xy}f(a, b) \\ D_{yx}f(a, b) & D_{yy}f(a, b) \end{pmatrix}$$

Exemple 10.4.4. Es considera la funció definida per:

$$f(x, y) = 2x^3y - xy^2 + e^{xy}.$$

Derivades parcials de primer ordre:

```
(%i1) f(x,y):=2*x^3*y - x*y^2 + exp(x*y);
(%o1) f(x,y):=2 x^3 y-x y^2+exp(x y)

(%i2) define (Dxf(x,y) , diff(f(x,y), x) );
      define (Dyf(x,y) , diff(f(x,y), y) );
(%o2) Dxf(x,y):=y %e^x y -y^2+6 x^2 y
      (%o3) Dyf(x,y):=x %e^x y -2 x y+2 x^3
```

Derivades parcials de segon ordre:

```
(%i4) define (Dxxf(x,y) , diff(Dxf(x,y), x) );
      define (Dxyf(x,y) , diff(Dxf(x,y), y) );
(%o4) Dxxf(x,y):= y^2 %e^x y + 12 x y
      (%o5) Dxyf(x,y):= x y %e^x y + %e^x y - 2 y + 6 :
(%i6) define (Dyxf(x,y) , diff(Dyf(x,y), x) );
      define (Dyyf(x,y) , diff(Dyf(x,y), y) );
(%o6) Dyxf(x,y):= x y %e^x y + %e^x y - 2 y + 6 x^2
      (%o7) Dyyf(x,y):= x^2 %e^x y - 2 x
```

Per calcular la matriu hessiana es pot construir una matriu amb les derivades de segon ordre calculades, o bé fer servir la instrucció “hessian” de wxMaxima que permet obtenir aquesta matriu directament:

```
(%i8) Hf(x,y) = matrix ( [Dxxf(x,y), Dxyf(x,y)] , [Dyxf(x,y), Dyyf(x,y)] ) ;
(%o8) Hf(x,y)=
      y^2 %e^x y + 12 x y      x y %e^x y + %e^x y - 2 y + 6 x^2
      x y %e^x y + %e^x y - 2 y + 6 x^2      x^2 %e^x y - 2 x
(%i9) Hf(x,y) = Hf: hessian (f(x,y) , [x,y]) ;
(%o9) Hf(x,y)=
      y^2 %e^x y + 12 x y      x y %e^x y + %e^x y - 2 y + 6 x^2
      x y %e^x y + %e^x y - 2 y + 6 x^2      x^2 %e^x y - 2 x
```

Per calcular la matriu hessiana en un punt, la sintaxi de wxMaxima és la següent:

```
(%i10) 'Hf(1,0) = Hf10:subst( [x=1, y=0] , Hf);
(%o10) Hf(1,0)=
      0  7
      7 -1
```

10.4.4. Derivades parcials de funcions reals de diverses variables

Per a funcions reals de més de dues variables la metodologia de càlcul de derivades parcials de primer ordre, de derivades direccionals i derivades parcials d'ordre superior és anàloga al cas de dues variables. Veurem la sintaxi mitjançant un exemple.

Exemple 10.4.5. Es considera la funció definida per:

$$f(x, y, z) = 4x^2y^2 + 3xe^{z^2}.$$

Definició de la funció:

```
(%i1) f(x,y,z):=4*x^2*y^2 + 3*x*exp(z^2);
(%o1) f(x,y,z):= 4 x2 y2 + 3 x exp(z2)
```

Derivades parcials de primer ordre:

```
(%i2) define (Dxf(x,y,z) , diff(f(x,y,z), x) );
      define (Dyf(x,y,z) , diff(f(x,y,z), y) );
      define (Dzf(x,y,z) , diff(f(x,y,z), z) );
(%o2) Dxf(x,y,z):= 3 %ez2 + 8 x y2
(%o3) Dyf(x,y,z):= 8 x2 y
(%o4) Dzf(x,y,z):= 6 x z %ez2
```

Derivades parcials en un punt:

```
(%i5) 'Dxf(1,2,-1) = Dxf(1,2,-1);
      'Dyf(1,2,-1) = Dyf(1,2,-1);
      'Dzf(1,2,-1) = Dzf(1,2,-1);
(%o5) Dxf(1,2,-1)= 3 %e + 32
(%o6) Dyf(1,2,-1)= 16
(%o7) Dzf(1,2,-1)= -6 %e
```

Derivades parcials de segon ordre:

```
(%i8) define (Dxxf(x,y,z) , diff(Dxf(x,y,z), x) );
      define (Dxyf(x,y,z) , diff(Dxf(x,y,z), y) );
      define (Dxzf(x,y,z) , diff(Dxf(x,y,z), z) );
(%o8) Dxxf(x,y,z):= 8 y2
(%o9) Dxyf(x,y,z):= 16 x y
(%o10) Dxzf(x,y,z):= 6 z %ez2
(%i11) define (Dyxf(x,y,z) , diff(Dyf(x,y,z), x) );
      define (Dyyf(x,y,z) , diff(Dyf(x,y,z), y) );
      define (Dyzf(x,y,z) , diff(Dyf(x,y,z), z) );
(%o11) Dyxf(x,y,z):= 16 x y
(%o12) Dyyf(x,y,z):= 8 x2
(%o13) Dyzf(x,y,z):= 0
```

```
(%i14) define (Dzxf(x,y,z) , diff(Dzf(x,y,z), x) );
      define (Dzyf(x,y,z) , diff(Dzf(x,y,z), y) );
      define (Dzzf(x,y,z) , diff(Dzf(x,y,z), z) );

(%o14) Dzxf(x,y,z):= 6 z %ez2
(%o15) Dzyf(x,y,z):= 0
(%o16) Dzzf(x,y,z):= 12 x z2 %ez2 + 6 x %ez2
```

Matriu hessiana:

```
(%i17) Hf(x,y,z) = matrix( [ Dxxf(x,y,z), Dxyf(x,y,z), Dxzf(x,y,z)] ,
                          [ Dyxf(x,y,z), Dyyf(x,y,z), Dyzf(x,y,z)] ,
                          [ Dzxf(x,y,z), Dzyf(x,y,z), Dzzf(x,y,z)] );

(%o17) Hf(x,y,z)=

$$\begin{bmatrix} 8y^2 & 16xy & 6ze^{z^2} \\ 16xy & 8x^2 & 0 \\ 6ze^{z^2} & 0 & 12xz^2e^{z^2} + 6xe^{z^2} \end{bmatrix}$$


(%i18) Hf(x,y) = Hf: hessian (f(x,y,z) , [x,y,z]) ;

(%o18) Hf(x,y)=

$$\begin{bmatrix} 8y^2 & 16xy & 6ze^{z^2} \\ 16xy & 8x^2 & 0 \\ 6ze^{z^2} & 0 & 12xz^2e^{z^2} + 6xe^{z^2} \end{bmatrix}$$

```

Matriu hessiana en un punt:

```
(%i19) 'Hf(1,2,-1) = Hf10:subst( [x=1, y=2, z=-1] , Hf);

(%o19) Hf(1,2,-1)=

$$\begin{bmatrix} 32 & 32 & -6e \\ 32 & 8 & 0 \\ -6e & 0 & 18e \end{bmatrix}$$

```

10-5.- Matriu jacobiana. Diferenciabilitat. Regla de la cadena.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_10-5.wxm**.

10.5.1 La matriu jacobiana.

Si la funció $f: A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ és derivable parcialment en un punt $a \in \overset{\circ}{A}$ s'anomena matriu jacobiana de f en aquest punt la matriu fila de dues columnes definida per:

$$Jf(a) = (D_x f(a) \quad D_y f(a))$$

Exemple 10.5.1. Es considera la funció definida per

$$f(x, y) = 2x^2y - 3xy^3$$

La sintaxi per al càlcul de la matriu jacobiana en un punt qualsevol i en un punt concret és la següent:

```
(%i1) f(x,y):=2*x^2*y-3*x*y^3;  
(%o1) f(x,y):= 2 x^2 y-3 x y^3  
  
(%i2) 'Jf(x,y) = Jf: jacobian([f(x,y)],[x,y]);  
(%o2) Jf(x,y)= [ 4 x y-3 y^3  2 x^2-9 x y^2]  
  
(%i3) 'Jf(1,2) = subst([x=1,y=2], Jf);  
(%o3) Jf(1,2)= [-16 -34]
```

Si es considera una funció real de diverses variables $f: A \subset \mathbb{R}^n \rightarrow \mathbb{R}$ tal que és derivable parcialment en un punt $a \in \overset{\circ}{A}$ s'anomena matriu jacobiana de la funció en aquest punt la matriu fila de n columnes definida per:

$$Jf(a) = (D_1 f(a) \quad D_2 f(a) \quad \cdots \quad D_n f(a))$$

Exemple 10.5.2. Es considera la funció definida per

$$f(x, y, z, t) = 2x^2y - 3zy^3 + 2z^2 \sin(t)$$

La sintaxi per al càlcul de la matriu jacobiana en un punt qualsevol i en un punt concret és la següent:

```
(%i1) f(x,y,z,t):=2*x^2*y-3*z*y^3+2*z^2*sin(t);
(%o1) f(x,y,z,t):=2 x^2 y-3 z y^3+2 z^2 sin(t)

(%i2) 'Jf(x,y,z,t) = Jf: jacobian([f(x,y,z,t)], [x,y,z,t]);
(%o2) Jf(x,y,z,t)=[ 4 x y  2 x^2-9 y^2 z  4 sin(t) z-3 y^3  2 cos(t) z^2]

(%i3) 'Jf(1,2,0,1) = subst([x=1,y=2,z=0,t=1], Jf);
(%o3) Jf(1,2,0,1)=[ 8  2 -24  0]
```

Si es considera una funció vectorial de diverses variables $f: A \subset \mathbb{R}^n \rightarrow \mathbb{R}^m$ tal que les funcions components són derivables parcialment en un punt $a \in \overset{o}{A}$, s'anomena matriu jacobiana de f en el punt a , la matriu de m files i n columnes i coeficients reals:

$$Jf(a) = (D_k f_i(a))_{\substack{1 \leq i \leq m \\ 1 \leq k \leq n}} = \begin{pmatrix} D_1 f_1(a) & D_2 f_1(a) & \cdots & D_n f_1(a) \\ D_1 f_2(a) & D_2 f_2(a) & \cdots & D_n f_2(a) \\ \cdots & \cdots & \cdots & \cdots \\ D_1 f_m(a) & D_2 f_m(a) & \vdots & D_n f_m(a) \end{pmatrix}$$

Si $m = n$ la matriu jacobiana és quadrada i el seu determinat s'anomena jacobià de la funció en el punt a .

Exemple 10.5.3. Es considera la funció definida per

$$f(x, y) = (2x^2y - 3xy^3, e^{x+y}, xy - 2x^2y)$$

La sintaxi per al càlcul de la matriu jacobiana en un punt qualsevol i en un punt concret és la següent:

```
(%i1) f1(x,y):=2*x^2*y-3*x*y^3;    f2(x,y):=exp(x+y);    f3(x,y):=x*y-2*x^2*y;
      f(x,y):= [f1(x,y) , f2(x,y) , f3(x,y)];
(%o1) f1(x,y):=2 x^2 y-3 x y^3
(%o2) f2(x,y):=exp(x+y)
(%o3) f3(x,y):=x y-2 x^2 y
(%o4) f(x,y):=[ f1(x,y), f2(x,y), f3(x,y)]
```

```
(%i5) 'Jf(x,y) = Jf: jacobian( f(x,y), [x,y]);
```

$$(\%o5) Jf(x,y) = \begin{bmatrix} 4xy - 3y^3 & 2x^2 - 9xy^2 \\ e^{y+x} & e^{y+x} \\ y - 4xy & x - 2x^2 \end{bmatrix}$$

```
(%i5) 'Jf(1,-1) = subst([x=1,y=-1], Jf);
```

$$(\%o5) Jf(1,-1) = \begin{bmatrix} -1 & -7 \\ 1 & 1 \\ 3 & -1 \end{bmatrix}$$

10.5.2. Diferenciabilitat

Considerem $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ una funció i un punt $a = (a_1, a_2) \in \overset{\circ}{A}$. Si existeix la matriu jacobiana de f en el punt a , $Jf(a) = (D_x f(a) \ D_y f(a))$, s'anomena diferencial de f en el punt a l'aplicació lineal $df_a : \mathbb{R}^2 \rightarrow \mathbb{R}$ tal que la seva matriu en les bases canòniques és $Jf(a)$, és a dir, tal que $[df_a]_{B_c B_c} = Jf(a)$. En virtut d'aquesta definició, per a un vector $h = (h_1, h_2) \in \mathbb{R}^2$ es compleix:

$$df_a(h) = Jf(a)[h] = (D_x f(a) \ D_y f(a)) \begin{pmatrix} h_1 \\ h_2 \end{pmatrix} = h_1 D_x f(a) + h_2 D_y f(a)$$

Es diu que $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ és diferenciable en el punt $a = (a_1, a_2) \in \overset{\circ}{A}$, si la funció df_a compleix la condició següent, anomenada condició de diferenciabilitat:

$$\lim_{h \rightarrow 0} \frac{f(a+h) - f(a) - df_a(h)}{|h|} = 0$$

és a dir:

$$\lim_{h \rightarrow 0} \frac{f(a_1 + h_1, a_2 + h_2) - f(a_1, a_2) - (h_1 D_x f(a) + h_2 D_y f(a))}{\sqrt{h_1^2 + h_2^2}} = 0$$

La condició de diferenciabilitat s'interpreta així: la funció diferencial df_a aproxima localment els increments de la funció. A més es l'única aplicació lineal que ho verifica.

L'existència de les derivades parcials és condició necessària per a la diferenciabilitat, però no és suficient.

Condicció suficient de diferenciabilitat. Si $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ és derivable parcialment i les derivades de primer ordre són funcions contínues en el punt $a \in \overset{\circ}{A}$ aleshores la funció és diferenciable en aquest punt.

Propietats de les funcions diferenciables en un punt. Si $f : A \subset \mathbb{R}^2 \rightarrow \mathbb{R}$ i $a \in \overset{\circ}{A}$, es compleixen les propietats següents:

- (1) Si la funció és diferenciable en el punt, aleshores és contínua en aquest punt (el recíproc no és necessàriament cert).
- (2) Si la funció és diferenciable en el punt, aleshores és derivable en qualsevol direcció i per a tot vector unitari $u \in \mathbb{R}^2$, es compleix $D_u f(a) = Jf(a)[u]$.
- (3) El recíproc de la propietat anterior no és cert, és a dir, una funció pot tenir derivada en qualsevol direcció en un punt i no ser diferenciable en aquest punt.

Exemple 10.5.4. Es considera la funció definida per

$$f(x, y) = x^2 + 2y^2$$

És clar que aquesta funció és diferenciable ja que compleix la condició suficient de diferenciabilitat. Per calcular la derivada en el punt $(3, 2)$ i en la direcció definida pel vector $u = (1, 1) \in \mathbb{R}^2$ en primer lloc es normalitzarà el vector, després es calcularà la matriu jacobiana de la funció en el punt i finalment es durà a terme el producte de la matriu jacobiana per la matriu columna associada al vector.

Vegem els càlculs amb wxMaxima:

```
(%i1) f(x,y):=x^2+2*y^2;
(%o1) f(x,y):=x^2+2*y^2

(%i2) 'u:=u:[1,1];    modul:sqrt(1^2+1^2);    'u1=u/modul;
(%o2) u=[1,1]
(%o3) sqrt(2)
(%o4) u1=[1/sqrt(2), 1/sqrt(2)]
```

```
(%i5) 'u1=u1 : matrix( [1/sqrt(2)] , [1/sqrt(2)] );
```

```
(%o5) u1 = 
$$\begin{bmatrix} \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} \end{bmatrix}$$

```

```
(%i6) 'Jf(x,y)=Jfxy : jacobian([f(x,y)], [x,y]);
```

```
(%o6) Jf(x,y) = 
$$\begin{bmatrix} 2x & 4y \end{bmatrix}$$

```

```
(%i7) 'Jf(3,2) = JfP : subst([x=3,y=2], Jfxy);
```

```
(%o7) Jf(3,2) = 
$$\begin{bmatrix} 6 & 8 \end{bmatrix}$$

```

```
(%i8) 'Duf(3,2)=JfP.u1;
```

```
(%o8) Duf(3,2) = 
$$2^{5/2} + 3\sqrt{2}$$

```

10.5.3 La regla de la cadena

Considerem dues funcions $f : A \subset \mathbb{R}^n \rightarrow \mathbb{R}^m$, $g : B \subset \mathbb{R}^m \rightarrow \mathbb{R}^p$ tals que:

- 1) la funció f és diferenciable en $a \in \overset{\circ}{A}$,
- 2) la funció g és diferenciable en $b = f(a) \in \overset{\circ}{B}$,
- 3) es compleix $f(A) \subset B$.

En aquestes condicions es pot considerar la funció composta

$$F = g \circ f : A \subset \mathbb{R}^n \rightarrow \mathbb{R}^p,$$

la qual compleix les propietats següents:

- 1) La funció F és diferenciable en el punt $a \in \overset{\circ}{A}$;
- 2) Es compleixen les igualtats:

$$dF_a = dg_b \circ df_a; \quad JF(a) = Jg(b)Jf(a)$$

Exemple 10.5.5. Es consideren les funcions diferenciables:

$$f(x, y) = (x^2 + y, xy^2, x + y^2), \quad g(x, y, z) = (e^{xy}, y + z).$$

Es tracta de calcular la matriu jacobiana de la funció composta $F = g \circ f$ en un punt qualsevol i calcular $JF(1, -1)$.

En primer lloc definim les funcions:

```
(%i1) f1(x,y):=x^2 + y$    f2(x,y):=x * y^2$    f3(x,y):=x + y^2$
      f(x,y):= [f1(x,y) , f2(x,y) , f3(x,y)];
(%o4) f(x,y):=[ f1(x,y), f2(x,y), f3(x,y)]

(%i5) g1(x,y,z):=%e^(x*y)$    g2(x,y,z):= y + z$
      g(x,y,z):= [g1(x,y,z) , g2(x,y,z) ];
(%o7) g(x,y,z):=[ g1(x,y,z), g2(x,y,z)]
```

Càlcul de les respectives matrius jacobianes:

```
(%i8) 'Jf(x,y)=Jfxy : jacobian(f(x,y), [x,y]);
(%o8) Jf(x,y)=

$$\begin{bmatrix} 2x & 1 \\ y^2 & 2xy \\ 1 & 2y \end{bmatrix}$$


(%i9) 'Jg(x,y,z)=Jgxyz : jacobian(g(x,y,z), [x,y,z]);
(%o9) Jg(x,y,z)=

$$\begin{bmatrix} y e^{xy} & x e^{xy} & 0 \\ 0 & 1 & 1 \end{bmatrix}$$

```

La matriu jacobiana de la funció composta en un punt qualsevol es calcula com el producte de matrius:

```
(%i10) 'JF(x,y) = Jgxyz.Jfxy;
(%o10) JF(x,y)=

$$\begin{bmatrix} x y^2 e^{xy} + 2 x y e^{xy} & 2 x^2 y e^{xy} + y e^{xy} \\ y^2 + 1 & 2 x y + 2 y \end{bmatrix}$$

```

Finalment, en el punt (1, -1):

```
(%i11) 'Jf(1,-1) = Jfa : subst([x=1,y=-1], Jfxy);
(%o11) Jf(1,-1)=

$$\begin{bmatrix} 2 & 1 \\ 1 & -2 \\ 1 & -2 \end{bmatrix}$$

```

10.5.4 Derivació de funcions compostes.

En el cas general de la regla de la cadena, si es compleix $m = 1$, la funció composta és una funció real. Aquest fet permet el càlcul de les derivades de la funció composta aplicant la regla de la cadena. Veurem a continuació tres exemples que il·lustren els possibles casos.

En primer lloc el cas $n = 1, m > 1$, que és la composició d'una funció vectorial de variable real i una funció real de variable vectorial.

Exemple 10.5.6. Es consideren les funcions definides per

$$f(t) = (2t + 1, t^2 + 2, \sin t) \quad , \quad g(x, y, z) = xy^2 + e^z$$

En primer lloc definim les funcions i la funció composta:

```
(%i1) f1(t):=2*t + 1;    f2(t):= t^2+2;    f3(t):=sin(t);  
      g(x,y,z):=x*y^2 + exp(z);  
(%o1) f1(t):= 2 t + 1  
(%o2) f2(t):= t^2 + 2  
(%o3) f3(t):= sin(t)  
(%o4) g(x, y, z):= x y^2 + exp(z)  
  
(%i5) F(t)=g(f1(t), f2(t), f3(t));  
(%o5) F(t)= %esin(t) + (2 t + 1)(t^2 + 2)^2
```

Ara podem calcular la derivada de la funció composta:

```
(%i6) DF(t)=diff(g(f1(t), f2(t), f3(t)) , t);  
(%o6) DF(t)= %esin(t) cos(t) + 2 (t^2 + 2)^2 + 4 t (2 t + 1)(t^2 + 2)  
  
(%i7) expand(%);  
(%o7) DF(t)= %esin(t) cos(t) + 10 t^4 + 4 t^3 + 24 t^2 + 8 t + 8
```

En segon lloc veurem un exemple del cas $n > 1, m = 1$, que és la composició d'una funció real de variable vectorial i una funció real de variable real.

Exemple 10.5.7. Es consideren les funcions definides per

$$f(x, y, z) = \sqrt{xy^2 + e^z} \quad , \quad g(t) = \sin t$$

En primer lloc definim les funcions i la funció composta:

```
(%i1) g(t):=sin(t);
      f(x,y,z):=sqrt( x*y^2 + exp(z)) ;
(%o1) g(t):=sin(t)
      (%o2) f(x,y,z):=sqrt(x y^2 + exp(z))

(%i3) F(x,y,z)=g(f(x,y,z));
(%o3) F(x,y,z)=sin(sqrt(%e^z + x y^2))
```

Les derivades parcials de la funció composta es calculen de la manera següent:

```
(%i4) Dx F(x,y,z)=diff(g(f(x,y,z)) , x);
(%o4) Dx F(x,y,z)=
$$\frac{y^2 \cos(\sqrt{e^z + x y^2})}{2 \sqrt{e^z + x y^2}}$$


(%i5) Dy F(x,y,z)=diff(g(f(x,y,z)) , y);
(%o5) Dy F(x,y,z)=
$$\frac{x y \cos(\sqrt{e^z + x y^2})}{\sqrt{e^z + x y^2}}$$


(%i6) Dz F(x,y,z)=diff(g(f(x,y,z)) , z);
(%o6) Dz F(x,y,z)=
$$\frac{e^z \cos(\sqrt{e^z + x y^2})}{2 \sqrt{e^z + x y^2}}$$

```

En tercer lloc veurem un exemple del cas $n > 1$, $m > 1$, que és la composició d'una funció vectorial de variable vectorial i una funció real de variable vectorial.

Exemple 10.5.8. Es consideren les funcions definides per

$$f(x, y) = (x + y, x - y, 2xy) \quad , \quad g(u, v, w) = \log(u^2 + v^2 - 2w)$$

En primer lloc definim les funcions i la funció composta:

```

(%i1) f1(x,y):=x+y;    f2(x,y):= x-y;    f3(x,y):=2*x*y;
      g(u,v,w):=log(u^2 + v^2 - 2*w) ;
(%o1) f1(x,y):=x+y
      (%o2) f2(x,y):=x-y
      (%o3) f3(x,y):=2 x y
      (%o4) g(u,v,w):=log(u^2+v^2+(-2)w)

(%i5) F(x,y)=g(f1(x,y), f2(x,y), f3(x,y));
(%o5) F(x,y)=log((y+x)^2-4 x y+(x-y)^2)

```

Les derivades parcials de la funció composta es calculen de la manera següent:

```

(%i6) Dx F(x,y)=diff(g(f1(x,y), f2(x,y), f3(x,y)) , x);
(%o6) Dx F(x,y)=
$$\frac{2(y+x)-4y+2(x-y)}{(y+x)^2-4xy+(x-y)^2}$$


(%i7) ratsimp(%);
(%o7) Dx F(x,y)=-
$$\frac{2}{y-x}$$


(%i8) Dy F(x,y)=diff(g(f1(x,y), f2(x,y), f3(x,y)) , y);
(%o8) Dy F(x,y)=
$$\frac{2(y+x)-2(x-y)-4x}{(y+x)^2-4xy+(x-y)^2}$$


(%i9) ratsimp(%);
(%o9) Dy F(x,y)=
$$\frac{2}{y-x}$$


```